
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 22.4.7 which is part of the NBIOT test suite in the ‘iwd-TTCN3-B2016-06_D17wk10’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 

Contact:
Ramakrishnan Thiruvathirai

ramakrishnan.thiruvathirai@rohde-schwarz.com

Tel. +441252666270
1.1 Verification Test Summary

Test Case:
TC_22_4_7
ATS Version:
iwd-TTCN3-B2016-06_D17wk10
System Simulator used:
CMW500
UE used:
Qualcomm MDM9206
Verification Status:
PASS
2. Corrections required

2.1 Change 1 – Correction to function ‘f_TC_22_4_7_NBIOT’

	Function name
	f_TC_22_4_7_NBIOT()

	Reason for change
	1. The TTCN does not implement Step 0A of the Prose moving directly to Step 0B after Preamble.
2. According to TS24.301 secn 6.3.2, the SS shall set the PTI= PTI unassigned(0) in case of user data transport via control plane.
3. According to TS 24.301 secn 5.3.1.2, If the UE receives the "Extended wait time" from the lower layers when no attach, tracking area updating or service request procedure is ongoing, the UE shall ignore the "Extended wait time". Hence, the TP1 mentioned in ts 36.523-1 secn 22.4.7.1 cannot be verified. Accordingly, TTCN implementation has been changed.
NOTE: A Prose CR shall be raised at RAN5#75

4. The TTCN CR R5s170350 introduces a flag p_Delay for the function f_NBIOT_RRC_ConnectionRelease. The RRC Connection Release sent at Step 3E should have the p_Delay flag set to FALSE so that there is no delay of 10s between Step 3E and Step 5

	Summary of change
	1. Implemented Step 0A of the Prose in TTCN
2. Used tsc_PTI_Unassigned for transmission of user data via control plane

3. Modified TTCN implementation by introducing a timer t_UplinkDataDelay and removal of steps 2A and 3A and enhancing the alt steps 3Aa1 to 3Ab1 as well as modifying the TTCN implementation between Steps 5 and Step 6Bb1
4. The flag p_Delay is set to FALSE.

	TTCN module
	NBIOT_RRC_ControlPlane.ttcn

	MCC160 Comment
	


Before change

  function f_TC_22_4_7_NBIOT() runs on NBIOT_PTC

  {//NB-IoT / RRC connection release / Success with extendedWait / RRC connection establishment / Reject with extendedWait

    var integer v_XtdWaitTime := 60;

    var float v_WaitForConnection := f_NBIOT_SetTimerToleranceMax (rrcTimer, int2float(v_XtdWaitTime));

    var boolean v_Rxd_RRCConnReq := false;

    var octetstring v_UserData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());

    var ProcedureTransactionIdentifier v_PTI;

    timer t_Wait;

    f_NBIOT_Init(c1);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE2A_NB_TESTLOOP_ModeG);

    f_NBIOT_CloseUE_TestLoopModeGH(nbiot_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 1, 8);

    f_NBIOT_TestBody_Set(true);

    v_PTI := f_NBIOT_MobileInfo_GetPTI(f_NBIOT_MobileInfo_GetNoOfPDNsConfigured ()-1);

    //@siclog "Step 0A" siclog@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, v_PTI, cs_UserDataContainer(v_UserData)))));

    //@siclog "Step 0B" siclog@

    //Wait for 1 s after the IP packet has been transmitted in step 0A.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    //@siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionRelease-NB message on Ncell 1 with IE extendedwaitTime set to 60s.

    f_NBIOT_RRC_ConnectionRelease_Common(nbiot_Cell1, tsc_SRB1bis, cs_RRCConnectionRelease_NB(tsc_RRC_TI_Def, other, -, v_XtdWaitTime), cs_TimingInfo_Now);

    //@siclog "Step 2" siclog@

    //Void.

    //@siclog "Step 2A" siclog@

    //Check: Does the UE transmit an RRCConnectionRequest-NB message on Ncell 1 within 60s?

    v_Rxd_RRCConnReq := f_NBIOT_RRC_RRCConnectionRequest_Check (nbiot_Cell1, v_WaitForConnection);

    if (v_Rxd_RRCConnReq) {

      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 2A");

    }

    //@siclog "Step 3" siclog@

    //Void.

    //@siclog "Step 3A" siclog@

    //The SS starts timer Timer_1 = 8 s

    t_Wait.start(8.0);

    //EXCEPTION: Steps 3Aa1 to 3Ab1 describe a transaction that depends on the UE behaviour; the "lower case letter" identifies a test sequence that takes place if a specific behaviour happens.

    //Note 2: A UE may send the pending loopback data

    alt {

      //@siclog "Step 3Aa1 - 3Aa4" siclog@

      //Steps 1 to 4 of the 'Generic Test Procedure NB-IoT Control Plan CIoT MO user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.3.3  are performed.

      //NOTE: The UE will transmit one ESM DATA TRANSPORT message containing loopback data received in step 0A.

      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(mo_Data)))){

        t_Wait.stop;

        f_NBIOT_508Check_CP_MONonSMSDataTransfer(nbiot_Cell1, cr_UserDataContainer);

        //@siclog "Step 3Aa5" siclog@

        //The SS transmits an RRCConnectionRelease-NB message

        f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

      }

      //@siclog "Step 3Ab1" siclog@

      //The SS waits for Timer_1 expiry

      [] t_Wait.timeout{}

    }

    //@siclog "Step 3B" siclog@

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3  are performed.

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 3C" siclog@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, v_PTI, cs_UserDataContainer(v_UserData)))));

    //@siclog "Step 3D" siclog@

    //Wait for 1 s after the IP packet has been transmitted.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    //@siclog "Step 3E" siclog@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

    //@siclog "Step 4" siclog@

    //Void.

    //@siclog "Step 5" siclog@

    //UE transmits an RRCConnectionRequest-NB message

    f_NBIOT_RRC_ConnectionRequest_Def(nbiot_Cell1);

    //@siclog "Step 6" siclog@

    //The SS responds with RRCConnectionReject-NB message with IE extendedwaitTime set to 60s.

    SRB.send(cas_NB_SRB0_RrcPdu_REQ(nbiot_Cell1, cs_TimingInfo_Now, cs_RRCConnectionReject_NB(v_XtdWaitTime)));

    //@siclog "Step 6A" siclog@

    //Check: Does the UE transmit an RRCConnectionRequest-NB message on Ncell 1 within 60s?

    v_Rxd_RRCConnReq := f_NBIOT_RRC_RRCConnectionRequest_Check (nbiot_Cell1, v_WaitForConnection);

    if (v_Rxd_RRCConnReq) {

      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 6A");

    }

    //@siclog "Step 6B" siclog@

    //The SS starts timer Timer_1 = 8 s

    t_Wait.start(8.0);

    //EXCEPTION: Steps 6Ba1 to 6Bb1 describe a transaction that depends on the UE behaviour; the "lower case letter" identifies a test sequence that takes place if a specific behaviour happens.

    //Note 2: A UE may send the pending loopback data

    alt {

      //@siclog "Step 6Ba1 - 6Ba4" siclog@

      //Steps 1 to 4 of the 'Generic Test Procedure NB-IoT Control Plan CIoT MO user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.3.3  are performed.

      //NOTE: The UE will transmit one ESM DATA TRANSPORT message containing loopback data received in step 3C.

      [] SRB.check(receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(mo_Data)))){

        t_Wait.stop;

        f_NBIOT_508Check_CP_MONonSMSDataTransfer(nbiot_Cell1, cr_UserDataContainer);

        //@siclog "Step 6Ba5" siclog@

        //The SS transmits an RRCConnectionRelease-NB message

        f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

      }

      //@siclog "Step 6Bb1" siclog@

      //The SS waits for Timer_1 expiry

      [] t_Wait.timeout{}

    }

    //@siclog "Step 7 - 12b1" siclog@

    //Void.

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble ( nbiot_Cell1, CONTROL_PLANE, N1_IDLE );

  }
After change

function f_TC_22_4_7_NBIOT() runs on NBIOT_PTC

  {//NB-IoT / RRC connection release / Success with extendedWait / RRC connection establishment / Reject with extendedWait

    var integer v_XtdWaitTime := 60;

    var float v_WaitForConnection := f_NBIOT_SetTimerToleranceMax (rrcTimer, int2float(v_XtdWaitTime));

    var boolean v_Rxd_RRCConnReq := false;

    var octetstring v_UserData := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());

    var ProcedureTransactionIdentifier v_PTI;

    timer t_Wait;

    //WA#

    timer t_UplinkDataDelay :=  f_NBIOT_SetTimerToleranceMax(nasTimer, 8.0) ;
    f_NBIOT_Init(c1);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, CONTROL_PLANE);

    //Bring UE to initial state

    f_NBIOT_Preamble(nbiot_Cell1, CONTROL_PLANE, STATE2A_NB_TESTLOOP_ModeG);

    f_NBIOT_CloseUE_TestLoopModeGH(nbiot_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 1, 8);

    //WA# The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);
    f_NBIOT_TestBody_Set(true);

    v_PTI := f_NBIOT_MobileInfo_GetPTI(f_NBIOT_MobileInfo_GetNoOfPDNsConfigured ()-1);

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3  are performed.

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1); //WA#
    //@siclog "Step 0A" siclog@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, tsc_PTI_Unassigned, //WA#  as per TS 24.301 cl 6.3.2, 

                                                  cs_UserDataContainer(v_UserData)))));

    //@siclog "Step 0B" siclog@

    //Wait for 1 s after the IP packet has been transmitted in step 0A.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    t_UplinkDataDelay.start; //WA#
    //@siclog "Step 1" siclog@

    //The SS transmits an RRCConnectionRelease-NB message on Ncell 1 with IE extendedwaitTime set to 60s.

    f_NBIOT_RRC_ConnectionRelease_Common(nbiot_Cell1, tsc_SRB1bis, cs_RRCConnectionRelease_NB(tsc_RRC_TI_Def, other, -, v_XtdWaitTime), cs_TimingInfo_Now);

    alt {

      //@siclog "Step 3Aa1 - 3Aa4" siclog@

      //Steps 1 to 4 of the 'Generic Test Procedure NB-IoT Control Plan CIoT MO user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.3.3  are performed.

      //NOTE: The UE will transmit one ESM DATA TRANSPORT message containing loopback data received in step 0A.

      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(mo_Data))){

        t_UplinkDataDelay.stop; //WA#
        f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step1"); //WA#
        f_NBIOT_508Check_CP_MONonSMSDataTransfer(nbiot_Cell1, cr_UserDataContainer,-,true,true); //WA#
        //@siclog "Step 3Aa5" siclog@

        //The SS transmits an RRCConnectionRelease-NB message

        f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

      }

      //@siclog "Step 3Ab1" siclog@

      //The SS waits for Timer_1 expiry

      [] t_UplinkDataDelay.timeout{

         f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Uplink data delay timer expired. No RRC connection request with cause mo data received from the UE");

      }
    }

    //@siclog "Step 3B" siclog@

    //'Generic Test Procedure NB-IoT Control Plane CIoT MT user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.2.3  are performed.

    f_NBIOT_508Check_CP_ResponseToPagingForMTAccess(nbiot_Cell1);

    //@siclog "Step 3C" siclog@

    //The SS transmits one IP packet to the UE embedded in a  ESM DATA TRANSPORT and  DLInformationTransfer-NB

    SRB.send(cas_NB_SRB_NasPdu_REQ(nbiot_Cell1,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, 

                                                                          tsc_PTI_Unassigned, //WA# as per TS 24.301 cl 6.3.2, 

                                                                         cs_UserDataContainer(v_UserData)))));

    //@siclog "Step 3D" siclog@

    //Wait for 1 s after the IP packet has been transmitted.

    //Note 1: The 1 second delay is used to secure that the UE have received and forwarded the IP Packet transmitted by the SS to the UE test loop function before the RRCConnectionRelease-NB message is sent by the SS.

    f_Delay(1.0);

    t_UplinkDataDelay.start; //WA#
    //@siclog "Step 3E" siclog@

    //The SS transmits an RRCConnectionRelease-NB message

    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1,-,-,-,-,false); //WA#
    //@siclog "Step 4" siclog@

    //Void.
    //@siclog "Step 5" siclog@

    alt {

      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(cr_EstablishmentCause_NB_Any))){

        t_UplinkDataDelay.stop; //WA#
        SRB.send(cas_NB_SRB0_RrcPdu_REQ(nbiot_Cell1, cs_TimingInfo_Now, cs_RRCConnectionReject_NB(v_XtdWaitTime)));

      }

      //@siclog "Step 3Ab1" siclog@

      //The SS waits for Timer_1 expiry

      [] t_UplinkDataDelay.timeout{

         f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "No RRC connection request received from the UE within 8s");

      }
    }

    //@siclog "Step 6A" siclog@

    //Check: Does the UE transmit an RRCConnectionRequest-NB message on Ncell 1 within 60s?

    v_Rxd_RRCConnReq := f_NBIOT_RRC_RRCConnectionRequest_Check (nbiot_Cell1, v_WaitForConnection);

    if (v_Rxd_RRCConnReq) {

      f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 6A");

    }

    //@siclog "Step 6B" siclog@

    //The SS starts timer Timer_1 = 8 s

    t_Wait.start(8.0);

    //EXCEPTION: Steps 6Ba1 to 6Bb1 describe a transaction that depends on the UE behaviour; the "lower case letter" identifies a test sequence that takes place if a specific behaviour happens.

    //Note 2: A UE may send the pending loopback data

    alt {

      //@siclog "Step 6Ba1 - 6Ba4" siclog@

      //Steps 1 to 4 of the 'Generic Test Procedure NB-IoT Control Plan CIoT MO user data transfer non-SMS transport' as described in TS 36.508 [18], clause 8.1.5A.3.3  are performed.

      //NOTE: The UE will transmit one ESM DATA TRANSPORT message containing loopback data received in step 3C.

      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionRequest_NB(mo_Data))){

        t_Wait.stop;

        f_NBIOT_508Check_CP_MONonSMSDataTransfer(nbiot_Cell1, cr_UserDataContainer,-,true, true); //WA#
        //@siclog "Step 6Ba5" siclog@

        //The SS transmits an RRCConnectionRelease-NB message

        f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1);

      }

      //@siclog "Step 6Bb1" siclog@

      //The SS waits for Timer_1 expiry

      [] t_Wait.timeout{}

    }

    //@siclog "Step 7 - 12b1" siclog@

    //Void.

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble ( nbiot_Cell1, CONTROL_PLANE, N1_IDLE );

  }
2.2 Change 2 – Correction to function ‘f_NBIOT_508Check_CP_MONonSMSDataTransfer’

	Function name
	f_NBIOT_508Check_CP_MONonSMSDataTransfer

	Reason for change
	1. According to TS 36.508 section 8.1.5A.3.3, the steps 3a1 to 3a4 should be executed only if px_DoAttachWithoutPDN is set to TRUE. If this pixit is seto FALSE, then Step 3b1 should be executed. The TTCN implementation does not follow this procedure sending a ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST at Step 3a2 independent of whether px_DoAttachWithoutPDN is set to TRUE or FALSE and waiting to receive an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT at Step 3a3. This is incorrect.
2. In the case of 22.4.7, as the TC has already received an RRC CONNECTION REQUEST with cause as MO_DATA at Step 3Aa1, there is no need to again wait for an RRC Connection Request.
3. In the case of TC 22.4.7, according to the Prose, at Steps 3Aa1 – 3Aa4, the UE would transmit one ESM DATA TRANSPORT message containing loopback data. Hence, the template cr_CONTROL_PLANE_SERVICE_REQUEST should check whether data has been received or not.

	Summary of change
	1. Brought the TTCN code from 3a1 to 3a4 under the purview of the pixit px_DoAttachWithoutPDN.
2. Introduced a flag p_Rcvd_RRC_Conn_Req which is set to FALSE by default to ensure that the TTCN is not waiting for an second RRC CONNECTION REQUEST in the case of 22.4.7
3. Introduced a flag p_DatatobeChecked which is set to FALSE by default to ensure that the template checks the reception of the ESM DATA TRANSPORT. 

	TTCN module
	NBIOT_CommonProcedures.ttcn

	MCC160 Comment
	


Before change
function f_NBIOT_508Check_CP_MONonSMSDataTransfer(NBIOT_CellId_Type p_CellId,

                                                    template (present) UserDataContainer p_DataExpected,

                                                    template (present) EstablishmentCause_NB_r13 p_EstCause := mo_Data) runs on NBIOT_PTC

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var AccessPointName v_APN;

    var integer v_PDNIndex := 0;

    var template (value) PDN_Address v_PDN_Address;

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;

    }

    if (px_DoAttachWithoutPDN) {

      v_AttachWithoutPDN := true_;

      v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                        v_PDNType,

                                                                        -,

                                                                        v_HeaderCompressConfigRX);

      v_NasInd_Expected := (cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B)), v_PDNConnectivityReq_Expected), // with PDN Connectivity

                          cr_NAS_Indication (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B))));                // or without PDN Connectivity

    } else {

      v_NasInd_Expected := cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B)), cr_ESM_DATA_TRANSPORT (?, p_DataExpected));

    }

    // Step 1

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, p_EstCause); // @sic R5-171137 sic@

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step1");

    // Step 2

    f_NBIOT_RRC_ConnectionSetup_Def (p_CellId);

    // Step 3

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    if (ispresent (v_NasInd_Rxd.Pdu.PiggybackedPduList)) { // Step 3a1a1

      if (ischosen (v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST)) {

        v_PDNConnectivityReq_Rxd := v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg;

        f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1a1");

      } else {

        f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Unexpected Message at Step3a1a1");

      }

    } else { // Step 3a1b1

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1b1");

      // Step 3a1b2

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          v_PDNConnectivityReq_Expected))) -> value v_ReceivedAsp;

      v_PDNConnectivityReq_Rxd := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1b2");

    }

    v_PDNIndex := f_NBIOT_MobileInfo_GetNoOfPDNsConfigured();

    f_NBIOT_MobileInfo_SetAssignedPdnAndPTI (v_PDNIndex,

                                             v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue,

                                             v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier);

    v_PcoFromUE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions;

    v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE));

    if (isvalue(v_PcoFromUE)) {

        f_NBIOT_MobileInfo_SetExtPCO (v_PDNIndex, valueof(v_PcoFromUE));

    }

    if (ispresent (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName)) {

      v_APN := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName;

      v_APN.iei := omit;

    } else {

      v_APN := valueof(cs_AccessPointName(px_AccessPointName));

    }

    f_NBIOT_MobileInfo_SetAPN (v_PDNIndex, v_APN); // NOTE: APN is not omit anymore

    if (match (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, '101'B)) {

      v_PDN_Address := cs_PDN_Address_NonIP;

    } else {

      v_PDN_Address := f_NBIOT_GetPDNAddress(f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE));

    }

    // Step 3a2

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                            f_NBIOT_MobileInfo_GetPTI(v_PDNIndex),

                                                                                            v_APN,

                                                                                            v_PDN_Address,

                                                                                            cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),

                                                                                            -,

                                                                                            -,

                                                                                            -,

                                                                                            v_HeaderCompressConfigTX,

                                                                                            cs_ControlPlaneInd,

                                                                                            cs_ServingPLMNRateControl(int2oct (4, 2))))));      // Step 4a1b3a4

    // Step 3a3

    SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                      tsc_SRB1bis,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId))));

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a3");

    if (pc_IP_PDN) {

      f_NBIOT_DelayForUserPlaneSignalling_CP(p_CellId);

    }

    if (px_DoAttachWithoutPDN) {

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          cr_ESM_DATA_TRANSPORT (?, p_DataExpected))));

    }

    // Step 4

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_SERVICE_ACCEPT)));

  }
After change

function f_NBIOT_508Check_CP_MONonSMSDataTransfer(NBIOT_CellId_Type p_CellId,

                                                    template (present) UserDataContainer p_DataExpected,

                                                    template (present) EstablishmentCause_NB_r13 p_EstCause := mo_Data,

                                                    boolean p_Rcvd_RRC_Conn_Req := false,

                                                    boolean p_DatatobeChecked := false) runs on NBIOT_PTC

  {

    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);

    var template (present) NAS_MSG_Indication_Type v_NasInd_Expected;

    var NAS_MSG_Indication_Type v_NasInd_Rxd;

    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;

    var NB_SRB_COMMON_IND v_ReceivedAsp;

    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;

    var ExtdProtocolConfigOptions v_PcoToUE;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var AccessPointName v_APN;

    var integer v_PDNIndex := 0;

    var template (value) PDN_Address v_PDN_Address;

    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;

    var template (omit) HeaderCompressConfig v_HeaderCompressConfigTX := omit;

    var integer v_SelectedPlmnIndex := f_NBIOT_CellInfo_GetSelectedPlmnIndex(p_CellId);

    var template (omit) RRCConnSetupComplete_AttachWithoutPDN_Type v_AttachWithoutPDN := omit;

    if (pc_HCCPCIoT and pc_IP_PDN){

      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;

      v_HeaderCompressConfigTX := crs_HeaderCompressConfigZeros;

    }

    if (px_DoAttachWithoutPDN) {
      v_AttachWithoutPDN := true_;

      v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                        v_PDNType,

                                                                        -,

                                                                        v_HeaderCompressConfigRX);

      v_NasInd_Expected := (cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B)), v_PDNConnectivityReq_Expected), // with PDN Connectivity

                          cr_NAS_Indication (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B))));                // or without PDN Connectivity

          // Step 1

    f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, p_EstCause); // @sic R5-171137 sic@

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step1");

    // Step 2

    f_NBIOT_RRC_ConnectionSetup_Def (p_CellId);

    // Step 3

    v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    if (ispresent (v_NasInd_Rxd.Pdu.PiggybackedPduList)) { // Step 3a1a1

      if (ischosen (v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST)) {

        v_PDNConnectivityReq_Rxd := v_NasInd_Rxd.Pdu.PiggybackedPduList[0].Msg;

        f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1a1");

      } else {

        f_NBIOT_SetVerdictFailOrInconc (__FILE__, __LINE__, "Unexpected Message at Step3a1a1");

      }

    } else { // Step 3a1b1

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1b1");

      // Step 3a1b2

      SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          v_PDNConnectivityReq_Expected))) -> value v_ReceivedAsp;

      v_PDNConnectivityReq_Rxd := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;

      f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a1b2");

    }

    v_PDNIndex := f_NBIOT_MobileInfo_GetNoOfPDNsConfigured();

    f_NBIOT_MobileInfo_SetAssignedPdnAndPTI (v_PDNIndex,

                                             v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue,

                                             v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier);

    v_PcoFromUE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions;

    v_PcoToUE := valueof(f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE));

    if (isvalue(v_PcoFromUE)) {

        f_NBIOT_MobileInfo_SetExtPCO (v_PDNIndex, valueof(v_PcoFromUE));

    }

    if (ispresent (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName)) {

      v_APN := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.accessPointName;

      v_APN.iei := omit;

    } else {

      v_APN := valueof(cs_AccessPointName(px_AccessPointName));

    }

    f_NBIOT_MobileInfo_SetAPN (v_PDNIndex, v_APN); // NOTE: APN is not omit anymore

    if (match (v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.pdnType.typeValue, '101'B)) {

      v_PDN_Address := cs_PDN_Address_NonIP;

    } else {

      v_PDN_Address := f_NBIOT_GetPDNAddress(f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE));

    }

    // Step 3a2

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                            f_NBIOT_MobileInfo_GetPTI(v_PDNIndex),

                                                                                            v_APN,

                                                                                            v_PDN_Address,

                                                                                            cs_ExtdPCO (int2oct(oct2int(v_PcoToUE.iel), 2), v_PcoToUE.pco),

                                                                                            -,

                                                                                            -,

                                                                                            -,

                                                                                            v_HeaderCompressConfigTX,

                                                                                            cs_ControlPlaneInd,

                                                                                            cs_ServingPLMNRateControl(int2oct (4, 2))))));      // Step 4a1b3a4

    // Step 3a3

    SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                      tsc_SRB1bis,

                                      cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cdr_CIOT_ActivateDefEpsBearerContextAccept(tsc_EpsDefaultBearerId))));

    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step3a3");

    if (pc_IP_PDN) {

      f_NBIOT_DelayForUserPlaneSignalling_CP(p_CellId);

    }

    SRB.receive(car_NB_SRB_NasPdu_IND(p_CellId,

                                        tsc_SRB1bis,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                          cr_ESM_DATA_TRANSPORT (?, p_DataExpected))));

    } else {

     if(p_DatatobeChecked)

     { 

       v_NasInd_Expected := cr_NAS_Indication(tsc_SHT_IntegrityProtected_PartiallyCiphered,cr_CONTROL_PLANE_SERVICE_REQUEST(?,cr_DataServiceType(-, '000'B), cr_ESM_MessageContainer));

     } else {

       v_NasInd_Expected := cr_NAS_Indication(tsc_SHT_IntegrityProtected_PartiallyCiphered,cr_CONTROL_PLANE_SERVICE_REQUEST(?,cr_DataServiceType(-, '000'B), omit)); 

     }

     if(p_Rcvd_RRC_Conn_Req == false)

     {

        f_NBIOT_RRC_ConnectionRequest_Def(p_CellId, p_EstCause); // @sic R5-171137 sic@

        f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "508Check CP MO NonSMS Data Transfer Step1");

     }

     // Step 2

     f_NBIOT_RRC_ConnectionSetup_Def (p_CellId);

     // Step 3

     v_NasInd_Rxd := f_NBIOT_RRCConnectionSetupComplete_Def(p_CellId,

                                                           cr_508_RRCConnectionSetupComplete_NB(tsc_RRC_TI_Def, v_SelectedPlmnIndex, -, -, -, v_AttachWithoutPDN),

                                                           v_NasInd_Expected);

    }
    // Step 4

    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,

                                   tsc_SRB1bis,

                                   cs_TimingInfo_Now,

                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                  cs_SERVICE_ACCEPT)));

  }
2.3 Change 3 – Correction to function ‘f_NBIOT_508Check_CP_MOSMSDataTransfer’

	Function name
	f_NBIOT_508Check_CP_MOSMSDataTransfer

	Reason for change
	As highlighted in previous change:

In the case of TC 22.4.7, according to the Prose, at Steps 3Aa1 – 3Aa4, the UE would transmit one ESM DATA TRANSPORT message containing loopback data. Hence, the template cr_CONTROL_PLANE_SERVICE_REQUEST would check whether data has been received or not. However, in this function, there is no need to check whether ESM MESSAGE CONTAINER is present or not.

	Summary of change
	Set the parameter p_EsmMessageContainer to omit.

	TTCN module
	NBIOT_CommonProcedures.ttcn

	MCC160 Comment
	


Before change
function f_NBIOT_508Check_CP_MOSMSDataTransfer(NBIOT_CellId_Type p_CellId, template (present) octetstring p_DataExpected) runs on NBIOT_PTC

{
   //SKIPPED CODE

   v_NasInd_Expected := (cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B), cr_NAS_MsgContainer_SMS(p_DataExpected)), v_PDNConnectivityReq_Expected), // with PDN Connectivity

                          cr_NAS_Indication (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B), cr_NAS_MsgContainer_SMS(p_DataExpected))));                // or without PDN Connectivity
  //SKIPPED CODE

}

After change
function f_NBIOT_508Check_CP_MOSMSDataTransfer(NBIOT_CellId_Type p_CellId, template (present) octetstring p_DataExpected) runs on NBIOT_PTC

{
   //SKIPPED CODE

   v_NasInd_Expected := (cr_NAS_IndicationWithPiggybacking (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B), omit, cr_NAS_MsgContainer_SMS(p_DataExpected)), v_PDNConnectivityReq_Expected), // with PDN Connectivity

                          cr_NAS_Indication (tsc_SHT_IntegrityProtected_PartiallyCiphered, cr_CONTROL_PLANE_SERVICE_REQUEST(?, cr_DataServiceType(-, '000'B), omit, cr_NAS_MsgContainer_SMS(p_DataExpected))));                // or without PDN Connectivity
  //SKIPPED CODE

}

2.4 Change 4 – Correction to template ‘cr_ESM_MessageContainer’

	Function name
	cr_ESM_MessageContainer

	Reason for change
	According to TS 24.301 section 9.9.3.15, the message can be coded as an LV or an TLV with the inclusion on the message container IEI. Hardcoding the iei to omit is incorrect

	Summary of change
	Expanded the scope of the iei to be anyoromit (*)

	TTCN module
	EPS_NAS_Templates.ttcn

	MCC160 Comment
	


Before change
template (present) ESM_MessageContainer cr_ESM_MessageContainer :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, POS) */

    iei := omit, // IE is mandatory in all MO messages

    iel := ?,

    esmPdu := ?

  };
After change
template (present) ESM_MessageContainer cr_ESM_MessageContainer :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NBIOT, POS) */

    iei := *, //WA#: omit, IE is mandatory in all MO messages
    iel := ?,

    esmPdu := ?

  };
2.5 Change 5 – Correction to template ‘cr_CONTROL_PLANE_SERVICE_REQUEST’

	Function name
	cr_CONTROL_PLANE_SERVICE_REQUEST

	Reason for change
	In the case of TC 22.4.7, according to the Prose, at Steps 3Aa1 – 3Aa4, the UE would transmit one ESM DATA TRANSPORT message containing loopback data. Hence, the template cr_CONTROL_PLANE_SERVICE_REQUEST should check whether data has been received or not

	Summary of change
	Introduced an parameter p_EsmMessageContainer which is set to omit by default to check the reception of an ESM DATA TRANSPORT message containing the loopback data in the case of 22.4.7

	TTCN module
	EPS_NAS_Templates.ttcn

	MCC160 Comment
	


Before change
template (present)  NAS_UL_Message_Type  cr_CONTROL_PLANE_SERVICE_REQUEST(template (present) NAS_KsiValue p_KsiValue,

                                                                            template (present) DataServiceType p_DataServiceType := cr_DataServiceType,

                                                                            template NAS_MessageContainer p_NASMsg := omit,

                                                                            template EPS_BearerContextStatus p_EPS_BearerContextStatus := cr_508_EPS_BearerContextStatusTAUReq ifpresent,

                                                                            template DeviceProperties p_DeviceProperties := cr_DeviceProperties('D'H) ifpresent) :=

  { // 24.301 cl. 8.2.33

    /* @status    APPROVED (NBIOT) */

    cONTROL_PLANE_SERVICE_REQUEST := {

      securityHeaderType        := tsc_SHT_NoSecurityProtection,                /* cl. 9.3.1    M V 1/2 */

      protocolDiscriminator     := tsc_PD_EMM,                                  /* cl. 9.2      M V 1/2 */

      messageType               := tsc_MT_ControlPlaneServiceRequest,           /* cl. 9.8      M V 1   */

      nasKeySetId               := cr_NAS_KeySetIdentifier (p_KsiValue, tsc_NasKsi_NativeSecurityContext),  /* cl. 9.9.3.21 M V 1/2 */

      dataServiceType           := p_DataServiceType,                           /* cl. 9.9.3.47 M V 1/2 */

      esmMessage                := cr_ESM_MessageContainer ifpresent,           /* cl. 9.9.3.15 O TLV-E 3-n IEI = 78 @sic R5s170032 sic@ */

      nasMessage                := p_NASMsg,                                    /* cl. 9.9.3.22 O TLV 4-253 IEI = 67 */

      epsBearerContextStatus    := p_EPS_BearerContextStatus,                   /* cl. 9.9.2.1  O TLV 4    IEI=0x57*/

      deviceProperties          := p_DeviceProperties                           /* cl. 9.9.2.0A O TV 1 IEI=0xD- */

    }

  };
After change
template (present)  NAS_UL_Message_Type  cr_CONTROL_PLANE_SERVICE_REQUEST(template (present) NAS_KsiValue p_KsiValue,

                                                                            template (present) DataServiceType p_DataServiceType := cr_DataServiceType,

                                                                            template (omit) ESM_MessageContainer p_EsmMessageContainer := omit, /*WA#*/
                                                                            template NAS_MessageContainer p_NASMsg := omit,

                                                                            template EPS_BearerContextStatus p_EPS_BearerContextStatus := cr_508_EPS_BearerContextStatusTAUReq ifpresent,

                                                                            template DeviceProperties p_DeviceProperties := cr_DeviceProperties('D'H) ifpresent) :=

  { // 24.301 cl. 8.2.33

    /* @status    APPROVED (NBIOT) */

    cONTROL_PLANE_SERVICE_REQUEST := {

      securityHeaderType        := tsc_SHT_NoSecurityProtection,                /* cl. 9.3.1    M V 1/2 */

      protocolDiscriminator     := tsc_PD_EMM,                                  /* cl. 9.2      M V 1/2 */

      messageType               := tsc_MT_ControlPlaneServiceRequest,           /* cl. 9.8      M V 1   */

      nasKeySetId               := cr_NAS_KeySetIdentifier (p_KsiValue, tsc_NasKsi_NativeSecurityContext),  /* cl. 9.9.3.21 M V 1/2 */

      dataServiceType           := p_DataServiceType,                           /* cl. 9.9.3.47 M V 1/2 */

      esmMessage                := p_EsmMessageContainer,           /* WA# cl. 9.9.3.15 O TLV-E 3-n IEI = 78 @sic R5s170032 sic@ */

      nasMessage                := p_NASMsg,                                    /* cl. 9.9.3.22 O TLV 4-253 IEI = 67 */

      epsBearerContextStatus    := p_EPS_BearerContextStatus,                   /* cl. 9.9.2.1  O TLV 4    IEI=0x57*/

      deviceProperties          := p_DeviceProperties                           /* cl. 9.9.2.0A O TV 1 IEI=0xD- */

    }

  };
3. Branches executed

None.

4. Execution Log Files

4.1 Qualcomm MDM9206
The Qualcomm MDM9206 passed this test case on Rohde & Schwarz CMW500 ConformanceTest System in LTE FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_22_4_7_MDM9206.log
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

5. References

	[1]
	R5s170xxx: Supporting information for agreement of NB-IoT test case 22.4.7. This archive comprises html and text format execution log files and PICS/PIXIT settings file respectively.


